
News 

Spike2 and Signal Training Courses 

Our U.K. training days will take place on the 20th – 21st of March at St. John’s Innovation 
Centre, Cambridge. This year the workshop thread will include sessions on using Signal 
for clamp-specific sampling applications, such as whole cell and single channel patch and 
voltage clamping experiments. Further details of the training days and a booking form are 
available from the training section of our website. 

The latest update for Spike2, 5.14, was released on Friday 27th January 2006 and is 
available from the CED downloads page for registered v5 users. This revision includes 
some new additions to the file importers.  

A full list of new features is available from the updates section of the website. 

A demonstration version of the current software is also available here.

Version 4.24 of Spike2 is downloadable for registered v4 users. 

Version 3.06 of Signal is now available for registered v3 users. This revision includes 
display modes for logarithmic axes. 

A demonstration version of Signal version 3 is available here.

Version 2.16 of Signal is downloadable for registered v2 users. 

Spike2 

Q. Can Spike2 read data from data loggers such as acceleration sensors? There is 
 no analogue output available but I do have a serial line feed from the sensor 
 giving X, Y and  Z co-ordinates. 

A. Spike2 can read data from the computers serial line (COM port) and display the 
 information as a TextMark. This special type of channel is normally used to 
 annotate the data record with text ‘notes’, which are more convenient and 
 easier to read than entering lines of text using the keyboard channel. When 
 recording these ‘notes’ from the serial line the incoming data is stored as a time 
 stamp and 4 bytes of marker information, followed by a text string.   

 The Source field in the Channel parameters dialog can be set to Manual or to a 
 serial port. You can set the terminator, baud rate, data bits and handshaking 
 once a serial port has been selected from the Source list.  

 When reading the serial line, the marker time is set when the first character 
 arrives, and the serial data for the current TextMark is read until a terminator 
 character is received or the serial string reaches 80 characters. TextMark data is 
 displayed as a small rectangle, the colours of which are determined by the first 
 marker code. Double click a TextMark to view and edit the text and codes. A list 
 of all TextMarks in the file and their associated data can be shown by clicking the 
 >> button in the Edit TextMark dialog. The data read from the accelerator sensor 
 gives X, Y and Z values as text as shown in the example below. 

  

 

  

   

 Edit TextMark dialog showing a list of available markers 

 

 TextMark data can also be viewed by changing the channel draw mode to State. 
 This displays the time between markers as a continuous ‘block’ with the option of 
 showing the associated text and marker codes as below. 

  
 It is quite easy to write a script to extract the X, Y and Z values. The attached 
 script, XYZval.s2s is  an example script that you could use as a template for 
 your requirements. 


'$XYZVal|Read text marks and creates X, Y and Z channels

'data from an accelerometer in the form of-

'X0455Y058Z0291M00T00000

'Data file must be open



Var data%;

Var X%, Y%, Z%;	'Channels to hold data

Var Chan%:=30;		'TextMark channel	







Data%:=frontview();	'Make sure data is at the front

if ViewKind(Data%) <> 0 then

Message("No data file found!\nPlease open a data file and run the script");

halt endif;



X%:=memchan(7,10);	

Chantitle$(X%,"X");

ChanShow(X%);

Y%:=memchan(7,10);

Chantitle$(Y%,"Y");

ChanShow(Y%);

Z%:=memchan(7,10);

Chantitle$(Z%,"Z");

ChanShow(Z%);			'Create memory channels for X, Y and Z



Doit%();

Optimise(X%,0,maxtime());

Optimise(Y%,0,maxtime());

Optimise(Z%,0,maxtime());



Func Doit%();		'This function does the work

Var Tim:=-1;

Var code%[4];

Var Data$;

Var XPos%, YPos%, Zpos%, Mpos%;

Var XPos$, YPos$, Zpos$;

Var XVal[1], Yval[1], Zval[1];

Repeat

 Tim:=Nexttime(Chan%,Tim,code%[],Data$);



	If tim>-1 then

	Cursor(1,TIM);

		XPos%:=InStr(Data$,"X");		'Find position of data in string

		YPos%:=InStr(Data$,"Y");

		ZPos%:=InStr(Data$,"Z");

		MPos%:=InStr(Data$,"M");

		If Xpos% and Ypos% and Zpos% and Mpos% >1 then

			Xpos$:=Mid$(Data$,XPos%+1,(Ypos%-1)-Xpos%);

			Ypos$:=Mid$(Data$,YPos%+1,(Zpos%-1)-Ypos%);

			Zpos$:=Mid$(Data$,ZPos%+1,(Mpos%-1)-Zpos%);

			Xval[0]:=Val(XPos$);

			Yval[0]:=Val(YPos$);

			Zval[0]:=Val(ZPos$);

			Memsetitem(X%,0, tim, Code%[],Xval[]);		'Copy the values

			Memsetitem(Y%,0, tim, Code%[],Yval[]);

			Memsetitem(Z%,0, tim, Code%[],Zval[]);



		Endif

	Endif

Until tim=-1;

Return 1

end

Q. Is it possible to display logarithmic axes in Signal? 

A. New additions to the X and Y range dialogs in Signal 3.06 allow the display of 
 logarithmic axis for data, memory and XY views. Checking the Logarithmic option 
 in the dialog switches the axis display from linear to logarithmic.  

 

X-Axis range dialog with new options for logarithmic axis display 

  

 If the logarithmic display mode is selected, this enables a second checkbox 
 option to Show powers. Select this to display the powers of the large tick spacing 
 unit on the axes, as shown on the X-axis below. 

 

Power spectrum displayed with logarithmic axes  

  

 The commands XAxisAttrib() and YAxisAttrib() are available to give script writers 
 control over logarithmic axes display in Signal. 

  

Back to top
Right-click the script icon 
and save to disk. 

If you have any problems 
opening the embedded 
scripts in this newsletter 
please let us know.  

Scripts: Spike2  
Q. Is there any way to create an average of multiple result views, such as PSTH’s or 
 waveform averages? 

A. The attached script, Grand Average.s2s, can be used to ‘build’ an average of 
 multiple result views in Spike2 including  waveform averages, PSTH’s and 
 interval histograms. You add result views into the Grand Average by opening a 
 saved result view or creating a new result view and making it the current or 
 ‘top’ window. Then simply click the ADD button in the script toolbar to add 
 it to the Grand Average.  


'|Grand average. Forms a new average from multiple averages. You must make the average to 'ADD' the current view.

'Not checking for errors in terms of different sized arrays or whether error bars are available in results.

 



Var result%;

Var Avg%;

Var Bins;

Var offset;

Var ok%;



Avg%:=frontview();



If ViewKind(avg%)<>4 then

   message("Please make the average the front view");

Halt

endif;



Bins:=binsize();

Bins:=XHigh()-Xlow();

Offset:=BinToX(Xlow());

offset:=Offset*-1;

'result%:=SetAverage(1,Bins,Offset,0,4);

result% := SetResult(1,Bins,View(avg%).BinSize(),View(avg%).XLow(),"Grand average","s",View(avg%).ChanUnits$(1),"Time",View(avg%).ChanTitle$(1),4);

DrawMode(1,2,0,1);

Sweeps(View(avg%).sweeps());



ArrAdd(View(result%).[],view(avg%).[]);

GetErrors();

Optimise(-1);

WindowTitle$("Grand Average");

Windowvisible(1);



repeat

ok%:=Interact("Select another average",2+4+8+16+32+64+128+256+512,0,"Quit","Add");

If ok%=1 then halt endif



Avg%:=Frontview();

ArrMul(View(avg%,1).[],View(avg%).Sweeps());								'Multiply new average to include by number of sweeps

ArrMul(View(result%,1).[],View(result%).Sweeps());						'Multiply result by number of sweeps to this point		

ArrAdd(View(result%).[],view(avg%).[]);									'Add new average to result

ArrDiv(View(avg%,1).[],View(avg%).Sweeps());								'Return new average values

View(Result%).Sweeps(Sweeps()+View(avg%).Sweeps());					'Update number of sweeps in result

ArrDiv(View(result%,1).[],View(result%).Sweeps());						'Divide result by number of sweeps

GetErrors();

Optimise(-1);

Until ok%=1;



Func GetErrors();

var Arr[bins];

Var old[bins];



View(avg%).BinError(1,0,arr[],0);

View(result%).BinError(1,0,Old[],0);

ArrAdd(Arr[],old[]);

View(result%).BinError(1,0,arr[],1);

Return 1

end

Scripts: Signal  Back to top
Right-click the script icon 
and save to disk. 

If you have any problems 
opening the embedded 
scripts in this newsletter 
please let us know. 

Q. Is it possible to ‘down-sample’ a Signal data file to a lower sampling rate?  

A. The attached script, Downsample.sgs, can be used to create a downsampled 
 duplicate view of a Signal data file. This creates a new view displaying one data 
 point in n of the original file. 

Did you know…? 
The Copy Spreadsheet command used in a time view now allows you to set any time 
interval for the copied data; previously it was constrained to integer multiples of the file 
time resolution.  

Recent questions  Back to top
Q. I'm getting myself very confused because of the channel view differences 
 between Signal and Spike2. Is there anyway to alter the  channel display in Spike 
 so that channel 0 is at the top (as in Signal) 

A. You can access the Edit: Preferences dialog in Spike2 and Signal to control the 
 settings for general preferences, display and sampling options among others. 
 These preferences are stored in the Windows registry and are specific to each   

 user login on the host PC. The display tab contains options for data display, 
 printing and image export. One of the options in the display tab in both  Spike2 
 and Signal is a checkbox to change the standard display to show the lowest 
 numbered channels at the top of a data file. You can also change the order 
 of channels in a view simply by clicking and dragging the channel numbers. 

 

 

 

 

 

 

 

Spike2 Edit preferences dialog 

User Group  
If you have any comments about the newsletter format and content, or wish to un-
subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk. 

mailto:sales@ced.co.uk

'$Downsample|Script to produce a new downsampled version of a Signal data file. 

var oldFile%;

View(App(3)).WindowVisible(0);





'---------Declare global variables-----------

var newFile%;									'Handle for downsampled data file

var oldSize;									'Bin size in source file

var newSize;									'Bin size in new file

var points%;									'Number of data points in source file frames

var newPoints%;									'Number of points in downsampled file

var factor%;									'Down samping factor	

var numChans%;									'Number of channels

var chan%,frame%;								'Counters for loops

var numFrames%;									'Number of frames





'---------Set toolbar buttons--------------

ToolbarSet(1,"Quit");

ToolbarSet(3,"Load source file",Load%);

ToolbarSet(5,"Downsample",DoDownSample%);

ToolbarSet(7,"Save",DoSave%);

ToolbarEnable(5,0);

ToolbarEnable(7,0);

Toolbar("Downsampling script",1023);





Func Load%()

if oldFile% > 0 then 							'If existing file opened

	View(oldFile%);

	FileClose(-1);								'Close existing file	

endif;

oldFile%:=FileOpen("",0,0);						'Open new source file

if oldFile% < 0 then							'Check file opened

	Message ("Failed to open file");

endif;

Window(0,0,100,100);							'Position window

FrontView(oldFile%);							'Bring to front

if oldFile% >0 then

	ToolbarEnable(5,1);

endif;

return 1;

end;





Func DoDownSample%()									

DlgCreate("Downsampling factor");				'Dialog to enter downsampling factor

DlgInteger(1,"Enter factor for down sampling",2,100);

DlgShow(factor%);

View(oldFile%);

points%:=Len(View(oldFile%,1).[]);				'Get number of data points in source file

numChans%:=ChanCount(-4);						'Count waveform channels

numFrames%:=FrameCount();							

newPoints%:=points%/factor%;					'Get number of data points for new file

oldSize:=Binsize(1);							'Get bin sizes for files

newSize:=oldSize * factor%;



newFile%:=SetMemory(numChans%,newPoints%,newSize,View(oldFile%).MinTime(),0,0,0,"Down sampled","s");	

												'Create empty data file with adjusted bin size and length 

Window(50,0,100,100);

DownSample();

return 1;

end;





Proc DownSample()								'Procedure to loop through frames and channels

var countF%;									'Counter to loop through each frame

var countC%;									'Counter to loop through each channel

for countF%:=1 to numFrames% do					'For each frame

	View(oldFile%).Frame(countF%);				'Ensure correct frame is viewed	

	View(newFile%).Frame(countF%);

	for countC%:=1 to numChans% do				'For each channel	

		DSChan(countC%);						'Link to procedure for downsampling channel

	next;

	if countF% < numFrames% then				'if not the last frame

		View(newFile%).AppendFrame();			'Append empty frame to new file	

	endif;

next;

View(oldFile%);Window(0,0,50,100);				'Position window

FrontView(oldFile%);					

FrontView(newFile%);

ToolbarEnable(7,1);

end;





Proc DSChan(chan%)

var count%;											'Counter to loop through data points

var num%:=0;

var oldDataArr[points%+factor%];					'Declare array to contain data from source channel	

ArrConst(oldDataArr[],View(oldFile%,chan%).[]);		'Put data into array

var newDataArr[newPoints%+1];						'Declare array for new data

'PrintLog("Old %d new %d\n", points%, newPoints%);	'For debug purposes

for count%:=0 to points% step factor% do			'Loop through source data every n points

    'PrintLog("Copy from %d to %d\n", count%, num%);	'For debug purposes

	newDataArr[num%]:=oldDataArr[count%];			'Put data into new array

	num%:=num% + 1;

next;

ArrConst(View(newFile%,chan%).[],newDataArr[]);		'Copy array to new file

View(newFile%).Optimise(-1);						'Optimise data 

end;



Func DoSave%()

var goodSave%;

View(newFile%);

goodSave%:=FileSaveAs("",0,"Save new file");

if not goodSave% then ToolbarEnable(7,0)endif;

return 1;

end;

'|Grand average. Forms a new average from multiple averages. You must make the average to 'ADD' the current view.

'Not checking for errors in terms of different sized arrays or whether error bars are available in results.

 



Var result%;

Var Avg%;

Var Bins;

Var offset;

Var ok%;



Avg%:=frontview();



If ViewKind(avg%)<>4 then

   message("Please make the average the front view");

Halt

endif;



Bins:=binsize();

Bins:=XHigh()-Xlow();

Offset:=BinToX(Xlow());

offset:=Offset*-1;

'result%:=SetAverage(1,Bins,Offset,0,4);

result% := SetResult(1,Bins,View(avg%).BinSize(),View(avg%).XLow(),"Grand average","s",View(avg%).ChanUnits$(1),"Time",View(avg%).ChanTitle$(1),4);

DrawMode(1,2,0,1);

Sweeps(View(avg%).sweeps());



ArrAdd(View(result%).[],view(avg%).[]);

GetErrors();

Optimise(-1);

WindowTitle$("Grand Average");

Windowvisible(1);



repeat

ok%:=Interact("Select another average",2+4+8+16+32+64+128+256+512,0,"Quit","Add");

If ok%=1 then halt endif



Avg%:=Frontview();

ArrMul(View(avg%,1).[],View(avg%).Sweeps());								'Multiply new average to include by number of sweeps

ArrMul(View(result%,1).[],View(result%).Sweeps());						'Multiply result by number of sweeps to this point		

ArrAdd(View(result%).[],view(avg%).[]);									'Add new average to result

ArrDiv(View(avg%,1).[],View(avg%).Sweeps());								'Return new average values

View(Result%).Sweeps(Sweeps()+View(avg%).Sweeps());					'Update number of sweeps in result

ArrDiv(View(result%,1).[],View(result%).Sweeps());						'Divide result by number of sweeps

GetErrors();

Optimise(-1);

Until ok%=1;



Func GetErrors();

var Arr[bins];

Var old[bins];



View(avg%).BinError(1,0,arr[],0);

View(result%).BinError(1,0,Old[],0);

ArrAdd(Arr[],old[]);

View(result%).BinError(1,0,arr[],1);

Return 1

end



